
LAMPIRAN 

#include <SPI.h> 

#include <MFRC522.h> 

#include <MD5Builder.h> 

#include <Wire.h> 

#include <LiquidCrystal_I2C.h> 

#include <Preferences.h> 

 

#define RST_PIN       4 

#define SS_PIN        5 

#define RELAY_PIN     25 

#define LCD_ADDR      0x27 

#define HASH_BLOCK    4 

 

#define max(a,b) ((a) > (b) ? (a) : (b)) 

MFRC522 mfrc522(SS_PIN, RST_PIN); 

MFRC522::MIFARE_Key key; 

LiquidCrystal_I2C lcd(LCD_ADDR, 16, 2); 

Preferences prefs; 

 

const String SALT = "SaltHashAgungPolsri2025";  

struct CardUser { 

  byte uid[4]; 

  uint32_t accessCount; 

}; 

 

CardUser allowedCards[] = { 

  { {0x33, 0x28, 0x17, 0x2D}, 0 },  

}; 

const int NUM_CARDS = sizeof(allowedCards) / sizeof(CardUser); 

void setup() { 

  Serial.begin(115200); 

  SPI.begin(); 

  mfrc522.PCD_Init(); 

 

  for (byte i = 0; i < 6; i++) key.keyByte[i] = 0xFF; 

  lcd.begin(); 

  lcd.backlight(); 

  lcd.setCursor(0, 0); 

  showReady(); 

 

  pinMode(RELAY_PIN, OUTPUT); 

  digitalWrite(RELAY_PIN, LOW); 

 

  // Load access counts from NVS 

  prefs.begin("rfid-acnt", false); 

  for (int i = 0; i < NUM_CARDS; i++) { 

    String key = "uid" + String(i); 

    allowedCards[i].accessCount = prefs.getUInt(key.c_str(), 0); 

    Serial.printf("[Init] UID %02X:%02X:%02X:%02X -> Access Count: %u\n", 

      allowedCards[i].uid[0], allowedCards[i].uid[1], 

      allowedCards[i].uid[2], allowedCards[i].uid[3], 

      allowedCards[i].accessCount); 

  } 

  prefs.end(); 

} 

 

void loop() { 

  if (!mfrc522.PICC_IsNewCardPresent() || !mfrc522.PICC_ReadCardSerial()) 

return; 

 

  lcd.clear(); 

  lcd.setCursor(0, 0); 

  lcd.print("UID: "); 

  displayUID(mfrc522.uid.uidByte); 

 

  CardUser* user = findUser(mfrc522.uid.uidByte); 

 

  if (!user) { 

    lcd.setCursor(0, 1); 

    lcd.print("Unauthorized"); 

    Serial.println("[Error] Unauthorized UID"); 

    delay(2000); 

    showReady(); 

    halt(); 

    return; 

  } 

 

  if (!authenticate(HASH_BLOCK)) { 

    lcd.setCursor(0, 1); 

    lcd.print("Auth Fail"); 

    Serial.println("[Error] Authentication to block failed"); 

    delay(2000); 

    showReady(); 

    halt(); 

    return; 

  } 

 

  if (user->accessCount == 0) { 

    Serial.println("[Info] First access. Storing initial hash."); 

    byte firstHash[16]; 

    generateHash(mfrc522.uid.uidByte, user->accessCount + 1, firstHash); 

    if (writeBlock(HASH_BLOCK, firstHash)) { 

      Serial.println("[OK] Initial hash written."); 

      Serial.print("New code generated: "); 

      for (int i = 0; i < 16; i++) Serial.printf("%02X ", firstHash[i]); 

      Serial.println(); 

      user->accessCount++; 

      saveAccessCount(user); 

    } else { 

      Serial.println("[Fail] Could not write initial hash."); 

    } 

    lcd.setCursor(0, 1); 

    lcd.print("Init Complete"); 

    grantAccess(); 

    delay(2000); 

    showReady(); 

    halt(); 

    return; 

  } 

 

  byte storedHash[18]; 

  if (!readBlock(HASH_BLOCK, storedHash)) { 

    lcd.setCursor(0, 1); 

    lcd.print("Read Fail"); 

    Serial.println("[Error] Read failed from card"); 

    delay(2000); 

    showReady(); 

    halt(); 

    return; 

  } 

 

  byte expectedHash[18]; 

  generateHash(mfrc522.uid.uidByte, user->accessCount, expectedHash); 

 

  Serial.print("UID: "); 

  for (int i = 0; i < 4; i++) Serial.printf("%02X ", mfrc522.uid.uidByte[i]); 

  Serial.println(); 

 

  Serial.print("Card Access Code: "); 

  for (int i = 0; i < 16; i++) Serial.printf("%02X ", storedHash[i]); 

  Serial.println(); 

 

  Serial.print("System Access Code: "); 

  for (int i = 0; i < 16; i++) Serial.printf("%02X ", expectedHash[i]); 

  Serial.println(); 

 

  if (memcmp(storedHash, expectedHash, 16) == 0) { 

    Serial.println("Code match, access granted."); 

    lcd.setCursor(0, 1); 

    lcd.print("Access Granted"); 

 

    user->accessCount++; 

 

    byte newHash[16]; 

    generateHash(mfrc522.uid.uidByte, user->accessCount, newHash); 

    Serial.print("New code generated: "); 

    for (int i = 0; i < 16; i++) Serial.printf("%02X ", newHash[i]); 

    Serial.println();n    

 

    if (writeBlock(HASH_BLOCK, newHash)) { 

      delay(50); 

      byte verifyHash[16]; 

      if (readBlock(HASH_BLOCK, verifyHash) && memcmp(newHash, 

verifyHash, 16) == 0) { 

        Serial.println("[OK] Hash updated and verified."); 

        saveAccessCount(user); 

        grantAccess(); 

      } else { 

        Serial.println("[Fail] Hash verification failed after write."); 

        lcd.setCursor(0, 1); 

        lcd.print("Hash Verify Fail"); 

      } 

    } else { 

      Serial.println("[Fail] Failed to write new hash."); 

      user->accessCount--; 

      lcd.setCursor(0, 1); 

      lcd.print("Write Error"); 

    } 

  } else { 

    Serial.println("Clone detected."); 

    lcd.setCursor(0, 1); 

    lcd.print("Clone Detected"); 

  } 

 

  Serial.println("w"); 

  yield(); 

  delay(2000); 

  yield(); 

  Serial.println("x"); 

  showReady(); 

  Serial.println("y"); 

  halt(); 

  Serial.println("z"); 

} 

 

void printUID(byte* uid) { 

  Serial.print("[UID] "); 

  for (byte i = 0; i < 4; i++) { 

    Serial.printf("%02X", uid[i]); 

    if (i < 3) Serial.print(":"); 

  } 

  Serial.println(); 

} 

 

void displayUID(byte* uid) { 

  for (byte i = 0; i < 4; i++) { 

    lcd.print(uid[i], HEX); 

    if (i < 3) lcd.print(":"); 

  } 

} 

 

CardUser* findUser(byte* uid) { 

  for (int i = 0; i < NUM_CARDS; i++) { 

    if (memcmp(uid, allowedCards[i].uid, 4) == 0) return &allowedCards[i]; 

  } 

  return nullptr; 

} 

 

bool authenticate(byte block) { 

  MFRC522::StatusCode status = mfrc522.PCD_Authenticate( 

    MFRC522::PICC_CMD_MF_AUTH_KEY_A, block, &key, &(mfrc522.uid)); 

  return status == MFRC522::STATUS_OK; 

} 

 

bool readBlock(byte block, byte* buffer) { 

  byte size = 18; 

  MFRC522::StatusCode status = mfrc522.MIFARE_Read(block, buffer, &size); 

  return status == MFRC522::STATUS_OK; 

} 

 

bool writeBlock(byte block, byte* data) { 

  MFRC522::StatusCode status = mfrc522.MIFARE_Write(block, data, 16); 

  return status == MFRC522::STATUS_OK; 

} 

 

void generateHash(byte* uid, uint32_t count, byte* output) { 

  String input = ""; 

 

  for (int i = 0; i < 4; i++) { 

    char hexPart[3]; 

    sprintf(hexPart, "%02X", uid[i]);   

    input += String(hexPart); 

  } 

 

  input += String(count); 

  input += SALT; 

 

  MD5Builder md5; 

  md5.begin(); 

  md5.add(input); 

  md5.calculate(); 

  md5.getBytes(output); 

} 

 

void grantAccess() { 

  digitalWrite(RELAY_PIN, HIGH); 

  delay(7000); 

  digitalWrite(RELAY_PIN, LOW); 

} 

 

void halt() { 

  mfrc522.PICC_HaltA(); 

  mfrc522.PCD_StopCrypto1(); 

  delay(2000); 

} 

 

void saveAccessCount(CardUser* user) { 

  prefs.begin("rfid-acnt", false); 

  for (int i = 0; i < NUM_CARDS; i++) { 

    if (memcmp(user->uid, allowedCards[i].uid, 4) == 0) { 

      char keyStr[8]; 

      snprintf(keyStr, sizeof(keyStr), "uid%d", i); 

      prefs.putUInt(keyStr, user->accessCount); 

      break; 

    } 

  } 

  prefs.end(); 

} 

 

void showReady() { 

  lcd.clear(); 

  printCentered(0, "Please tap your"); 

  printCentered(1, "card"); 

} 

 

void printCentered(int row, const String& message) { 

  int spaces = max((16 - message.length()) / 2, 0); 

  lcd.setCursor(spaces, row); 

  lcd.print(message); 

} 

